

NAG Fortran Library Routine Document

G07ABF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G07ABF computes a confidence interval for the mean parameter of the Poisson distribution.

2 Specification

```
SUBROUTINE G07ABF(N, XMEAN, CLEVEL, TL, TU, IFAIL)
INTEGER          N, IFAIL
real           XMEAN, CLEVEL, TL, TU
```

3 Description

Given a random sample of size n , denoted by x_1, x_2, \dots, x_n , from a Poisson distribution with probability function

$$p(x) = e^{-\theta} \frac{\theta^x}{x!}, \quad x = 0, 1, 2, \dots$$

the point estimate, $\hat{\theta}$, for θ is the sample mean, \bar{x} .

Given n and \bar{x} this routine computes a $100(1 - \alpha)\%$ confidence interval for the parameter θ , denoted by $[\theta_l, \theta_u]$, where α is in the interval $(0,1)$.

The lower and upper confidence limits are estimated by the solutions to the equations

$$e^{-n\theta_l} \sum_{x=T}^{\infty} \frac{(n\theta_l)^x}{x!} = \frac{\alpha}{2},$$

$$e^{-n\theta_u} \sum_{x=0}^T \frac{(n\theta_u)^x}{x!} = \frac{\alpha}{2},$$

where $T = \sum_{i=1}^n x_i = n\hat{\theta}$.

The relationship between the Poisson distribution and the χ^2 distribution (see Hastings and Peacock (1975), page 112) is used to derive the equations

$$\theta_l = \frac{1}{2n} \chi_{2T, \alpha/2}^2,$$

$$\theta_u = \frac{1}{2n} \chi_{2T+2, 1-\alpha/2}^2,$$

where $\chi_{\nu, p}^2$ is the deviate associated with the lower tail probability p of the χ^2 distribution with ν degrees of freedom.

In turn the relationship between the χ^2 distribution and the gamma distribution (see Hastings and Peacock (1975), page 70) yields the following equivalent equations;

$$\theta_l = \frac{1}{2n} \gamma_{T, 2; \alpha/2},$$

$$\theta_u = \frac{1}{2n} \gamma_{T+1, 2; 1-\alpha/2},$$

where $\gamma_{\alpha,\beta;\delta}$ is the deviate associated with the lower tail probability, δ , of the gamma distribution with shape parameter α and scale parameter β . These deviates are computed using G01FFF.

4 References

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

Snedecor G W and Cochran W G (1967) *Statistical Methods* Iowa State University Press

5 Parameters

- | | | |
|----|--|---------------------|
| 1: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the sample size, n . | |
| | <i>Constraint:</i> $N \geq 1$. | |
| 2: | XMEAN – <i>real</i> | <i>Input</i> |
| | <i>On entry:</i> the sample mean, \bar{x} . | |
| | <i>Constraint:</i> $XMEAN \geq 0.0$. | |
| 3: | CLEVEL – <i>real</i> | <i>Input</i> |
| | <i>On entry:</i> the confidence level, $(1 - \alpha)$, for two-sided interval estimate. For example CLEVEL = 0.95 gives a 95% confidence interval. | |
| | <i>Constraint:</i> $0.0 < CLEVEL < 1.0$. | |
| 4: | TL – <i>real</i> | <i>Output</i> |
| | <i>On exit:</i> the lower limit, θ_l , of the confidence interval. | |
| 5: | TU – <i>real</i> | <i>Output</i> |
| | <i>On exit:</i> the upper limit, θ_u , of the confidence interval. | |
| 6: | IFAIL – INTEGER | <i>Input/Output</i> |
| | <i>On entry:</i> IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details. | |
| | <i>On exit:</i> IFAIL = 0 unless the routine detects an error (see Section 6). | |
| | For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. When the value -1 or 1 is used it is essential to test the value of IFAIL on exit. | |

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 1$,
 or $XMEAN < 0.0$,
 or $CLEVEL \leq 0.0$,
 or $CLEVEL \geq 1.0$.

IFAIL = 2

When using the relationship with the gamma distribution to calculate one of the confidence limits, the series to calculate the gamma probabilities has failed to converge. Both TL and TU are set to zero. This is a very unlikely error exit and if it occurs please contact NAG.

7 Accuracy

For most cases the results should have a relative accuracy of $\max(0.5E-12, 50.0 \times \epsilon)$ where ϵ is the *machine precision* (see X02AJF). Thus on machines with sufficiently high precision the results should be accurate to 12 significant digits. Some accuracy may be lost when $\alpha/2$ or $1 - \alpha/2$ is very close to 0.0, which will occur if CLEVEL is very close to 1.0. This should not affect the usual confidence intervals used.

8 Further Comments

None.

9 Example

The following example reads in data showing the number of noxious weed seeds and the frequency with which that number occurred in 98 sub-samples of meadow grass. The data is taken from Snedecor and Cochran (1967), page 224. The sample mean is computed as the point estimate of the Poisson parameter θ . The routine G07ABF is then called to compute both a 95% and a 99% confidence interval for the parameter θ .

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G07ABF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real            CLEVEL, SUM, TL, TU, XMEAN
      INTEGER          I, IFAIL, IFREQ, N, NUM
*      .. External Subroutines ..
      EXTERNAL        G07ABF
*      .. Intrinsic Functions ..
      INTRINSIC       real
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G07ABF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)

*
*      Read in the number of Noxious Seeds in a sub sample and
*      the frequency with which that number occurs.
*
*      Compute the sample mean
*
      SUM = 0.0e0
      N = 0
20  READ (NIN,*,END=40) NUM, IFREQ
      SUM = SUM + real(NUM)*real(IFREQ)
      N = N + IFREQ
      GO TO 20
40  XMEAN = SUM/real(N)
      WRITE (NOUT,*)
      WRITE (NOUT,99999)
      + 'The point estimate of the Poisson parameter = ', XMEAN
*
```

```

      DO 60 I = 1, 2
        IF (I.EQ.1) THEN
          CLEVEL = 0.95e0
          WRITE (NOUT,*)
          WRITE (NOUT,*)
+         '95 percent Confidence Interval for the estimate '
        ELSE
          CLEVEL = 0.99e0
          WRITE (NOUT,*)
          WRITE (NOUT,*)
+         '99 percent Confidence Interval for the estimate '
        END IF
        IFAIL = 0
*
        CALL G07ABF(N,XMEAN,CLEVEL,TL,TU,IFAIL)
*
        WRITE (NOUT,99998) '( ', TL, ' , ', TU, ' )'
60 CONTINUE
    STOP
*
99999 FORMAT (1X,A,F6.4)
99998 FORMAT (6X,A,F6.4,A,F6.4,A)
    END

```

9.2 Program Data

G07ABF Example Program Data

0	3
1	17
2	26
3	16
4	18
5	9
6	3
7	5
8	0
9	1
10	0

9.3 Program Results

G07ABF Example Program Results

The point estimate of the Poisson parameter = 3.0204

95 percent Confidence Interval for the estimate
(2.6861 , 3.3848)

99 percent Confidence Interval for the estimate
(2.5874 , 3.5027)
